
INTRODUCTION 
Woodworkers who build bookshelves have to incorporate a notion of shelf sag into their design. 

Basically, given the type of shelf material, shelf load, load distribution, dimensions, and method of 
attachment it is possible to calculate the shelf load. Not all shelves are created equal, there are many 
factors which contribute to shelf deflection. For example a shelf made of pine will sag more than one 

made of hard maple. The shelf load may be increased by increasing the width of the board, increasing 
the thickness of the board, or shortening the shelf span.

PROGRAM SPECIFICATION
There are two types of shelves. The first has supported ends such as those which have adjustable 
shelving. 

  
s =

5wL4

384EI         (1)

where

 s = deflection

 w = load per unit length (lbs per inch)
 L = span length (inches)

 E = Young’s Modulus of Elasticity (psi)
 I = Moment of inertia (in4)

Secondly, a shelf with fixed ends, secured in housings.

  
s =

wL4

384EI         (2)

The (area) moment of inertia can be calculated from the breadth and thickness of the shelf board:

  
I =

(b · h3)

12         (3)

 b = breadth (depth) in inches
 h = thickness in inches

The modulus of elasticity (MOE) basically measures the stiffness of wood, and is a good indicator of a 

materials strength. MOE is expressed in pounds per square inch. Here are some MOE values for 
common woods (x 106):

 Ash   1.77
 Cherry    1.49

 Maple, sugar  1.83

 Oak, white   1.78
 Walnut, black  1.68

 Pine   1.30
 Poplar, yellow  1.58
 MDF   0.58

 Finnish birch plywood  1.48   (18mm, 13 ply)

Calculating the deflection in a shelf involves a considerable amount of user inputted information:

• shelf load per foot (lbs)

• shelf length, depth and thickness (in)

• modulus of elasticity (psi)

• whether the shelf is adjustable or fixed

From this input, two interim calculations are performed. Firstly, the uniform load lb/in is calculated, then 

the area moment of inertia is calculated. A target for allowable sag is 0.02 inches per foot or less.

THE ALGORITHM
There is nothing inherently complex about the algorithm for this program if the assumption is that all 

information about the shelf, and the material it is made of are provided by the user - information in, 
calculation out. 

1. Obtain user input for characteristics of the shelf (length, depth, thickness, and load per foot), and 
the modulus of elasticity of the wood being used to build the shelf.

2. Calculate the area moment of inertia, and shelf load per unit length.
3. Obtain user input as to whether the shelf is adjustable or fixed.

4. Calculate the shelf deflection.
5. Output the shelf deflection as a total shelf value, and also as a value per foot.
6. Determine if the shelf sag is acceptable.

THE PROGRAM SKELETON
The basic program looks like any program skeleton:

program shelfLoad

   implicit none

    ! declare variables

    ! input load per foot
    ! input shelf length, depth, thickness
    ! input modulus of elasticity

    ! input choice of fixed or adjustable shelf

    ! calculate area moment of inertia
    ! calculate shelf load per unit length
    
    ! input shelf fixed or adjustable

    ! calculate shelf deflection

    ! output shelf deflection: total value, value per foot

    ! determine whether sag is acceptable or not

end program shelfLoad

BUILDING THE PROGRAM 
One way of translating the algorithm into a program is to take each step in the algorithm and write the 
code for it. Once they have all been written, they can be integrated together. 

THE IDENTIFIERS
Firstly, what type of data does the program require from the user?

 Load per foot (pounds)
 Shelf length (inches)
 Shelf depth (inches)
 Shelf thickness (inches)
 Modulus of elasticity (psi)
 Choice of fixed or adjustable shelf

Next, what type of data is calculated within the program? It has to be stored somewhere as well.

 Load per unit length (lb/in)
 Area moment of inertia (inches4)
 Shelf deflection (inches)

 
As the calculations all involved real numbers, they should be declared as type double. The next step is to 

provide them with identifiers. The identifiers should be descriptive without being too long. Here are 
some suggestions:

 Load per foot (pounds)   loadFt, load_per_ft

 Shelf length (inches)   shlfLen, shelf_length, shlf_lngth

 Shelf depth (inches)   shlfDep, shelf_depth, shlf_dpth

 Shelf thickness (inches)   shlfThk, shelf_thickness, shlf_thck

 Modulus of elasticity (psi)  modElas, mod_Elasticity, mod_Elstcty

 Load per unit length (lb/in)  loadU, load_unit, loadShlf

 Area moment of inertia (inches4)  momentI, moment_inertia, momentInertia

 Shelf deflection (inches)   defltn, deflection, shlfSag

 Fixed or adjustable shelf  opt, option, shelf_type

As is shown, there are many different options. Selecting an identifier such as modulus_of_elasticity is 

extremely descriptive, but can be challenging from the perspective of using it in an equation. Even 
mod_Elasticity is somewhat long. modElas is probably the right length. Now the variables can be 
declared:

    program shelfLoad
 
       implicit none

       real :: loadFt, shlfLen, shlfDep, shlfThk
       real :: modElas, loadShlf, momentI, shlfSag
       integer :: opt

    end program shelfLoad

STEP 1.
Input the information needed to calculate the change in length.

   write (*,*) "Calculate Shelf Deflection (Sag)"

   write (*,fmt="(a21)",advance="no") "Load per foot (lbs): "
   read (*,*) loadFt
   write (*,fmt="(a19)",advance="no") "Shelf length (in): "
   read (*,*) shlfLen
   write (*,fmt="(a18)",advance="no") "Shelf depth (in): "
   read (*,*) shlfDep
   write (*,fmt="(a22)",advance="no") "Shelf thickness (in): "
   read (*,*) shlfThk
   write (*,fmt="(a29)",advance="no") "Modulus of Elasticity (psi): "
   read (*,*) modElas

All the write statements use an internal format statement, which in the case of these statements basically 
just specifies the ascii string in each of the prompts. The clause advance=”no”, stops write adding a 
newline at the end of the string.

STEP 2.
Perform the interim calculations. 

First calculate the uniform load in pounds/inch. This involves dividing the load by 12. The result is output 

to the user.

   loadShlf = loadFt / 12.0
   write (*,10) "Shelf load = ", loadShlf, " lb/in"
   10 format (a13, f5.2, a6)

Next calculate the area moment of inertia using Eq.(3):

I =
(b · h3)

12

becomes:

   momentI = (shlfDep * (shlfThk**3.0)) / 12.0;

Note the use of the ** operator to calculate a power of a number. It is also possible to write the equation 

like this:

   momentI = (shlfDep * (shlfThk * shlfThk * shlfThk)) / 12.0;

but it does seem messier. 

STEP 3.
Obtain information on whether or not the shelf is adjustable.

write (*,fmt="(a38)",advance="no") "(1) Adjustable or (2) Fixed shelving? "
read (*,*) opt

The read function uses an integer variable opt to hold the value, 1 for adjustable, 2 for fixed shelving.

STEP 4.
Calculate the shelf deflection.

As there are two possible equations, depending on whether the shelf is adjustable (Eq.1), or fixed (Eq.2), 
it is necessary to use an if statement to apply the appropriate equation based on the value of the 

variable opt.

   if (opt == 1) then
      shlfSag = 5.0 * loadShlf * shlfLen**4.0
   elseif (opt == 2) then
      shlfSag = loadShlf * shlfLen**4.0
   end if
   shlfSag = shlfSag / (384.0 * modElas * momentI)

In this case, the if statement deals with the numerator of each of Eqns. (1) and (2), and the denominator  

is applied afterwards. Note that the only real difference in these equations is the factor of 5 used in the 
scaling of the adjustable shelf. 

STEP 5.
Output the shelf deflection as a total shelf value.

   write (*,20) "Shelf deflection (sag) = ", shlfSag, " inches"
   20 format(a25, f5.3, a7)

STEP 6.
Determine if the shelf sag is acceptable.

   if (shlfSag/(shlfLen/12.0) < 0.02) then
      write (*,*) "Shelf sag is acceptable"
   else
      write (*,*) "Shelf sag is unacceptable"
   end if

TESTING THE PROGRAM 
Here is a test of the program using the same information from the sample run shown earlier.

Here is a sample run for a shelf (adjustable) made of sugar (hard) maple. The weight of 6 pounds 
represents 12 inches of paperback books. Measurements are in inches, because although Canada is 

“officially” metric,  lumber is still sold using imperial measures.

 Calculate Shelf Deflection (Sag)
Load per foot (lbs): 6
Shelf length (in): 28.5
Shelf depth (in): 5.5
Shelf thickness (in): 0.75
Modulus of Elasticity (psi): 1830000
Shelf load =  0.50 lb/in
(1) Adjustable or (2) Fixed shelving? 1
Shelf deflection (sag) = 0.012 inches
 Shelf sag is acceptable

EXTENDING THE PROGRAM 
The one problem with this program as it stands is requiring the user to enter a value for a piece of woods 

species’ “Modulus of elasticity”. A better idea would be integrate a series of wood species, and their 
associated “modulus of elasticity” (MOE) values directly into the program and allow the user to choose a 
wood species. To achieve this, we create two arrays, one to hold a series of string representing common 

wood species used to build shelves, and a second to hold the associated MOE values.

First we create an integer array to hold the MOE values, replacing the singular integer variable modElas 
with an array:

   integer, dimension(9) :: modElas = [1770000, 1490000, 1830000, &
             1780000, 1680000, 1300000, 1580000, 580000, 1480000]

In this case the array has been initialized during creation. Now we need a array of strings to hold the 

wood species names:

   character(len=15), dimension(9) :: woods

This creates an array of strings containing 9 strings, each of which can hold a name up to 15 characters in 

length. Now assigning values can be done in this manner (after all variable declarations):

   woods = [character(15) :: 'Ash','Cherry','Hard Maple','White Oak',&
            'Walnut','Pine','Poplar','MDF','Birch Plywood']

Then there just have to be some modifications to the underlying code base. The first of these is a piece 
of code to print the wood species menu, and allow the user to choose a wood species.

   write (*,fmt="(a23)") "Choose a wood species: "
   do i= 1,9
      write (*,fmt="(a3,i1,a2,a15)") "  (",i,") ",woods(i)
   end do
   write (*,fmt="(a2)",advance="no") "? "
   read (*,*) woodOpt

This produces a menu of the form:

Choose a wood species:
  (1) Ash
  (2) Cherry
  (3) Hard Maple
  (4) White Oak
  (5) Walnut
  (6) Pine
  (7) Poplar
  (8) MDF
  (9) Birch Plywood

Then we just have to modify the code related to the use of the array modElas. This means changing one 
line of code from:

   shlfSag = shlfSag / (384.0 * modElas * momentI)

to:

   shlfSag = shlfSag / (384.0 * modElas(woodOpt) * momentI)

The rest of the program remains the same.

Can the program be made even better? Likely - the program now has the capacity of nine wood species, 
but to make it even more useful, the program could be modified to take the wood species/MOE data 
from a file. This would mean that wood species could be added without affecting the program.

TRANSLATION TO JULIA
Now let’s translate the original program to Julia. There isn’t really much that changes in the code, just 

some of the control structures, and the removal of the variable declarations, as Julia is dynamically 
typed. A somewhat shorter program, mostly to do with variable declarations, and less complex println() 
statements. 

println("Calculate Shelf Deflection (Sag)")

println("Load per foot (lbs): ")
loadFt = parse(chomp(readline()))
println("Shelf length (in): ")
shlfLen = parse(chomp(readline()))
println("Shelf depth (in): ")
shlfDep = parse(chomp(readline()))
println("Shelf thickness (in): ")
shlfThk = parse(chomp(readline()))
println("Modulus of Elasticity (psi): ")
modElas = parse(chomp(readline()))

# Calculate uniform load lb/in
loadShlf = loadFt / 12.0
println("Shelf load = ", loadShlf, " lb/in")

# Calculate area moment of inertia
momentI = (shlfDep * (shlfThk^3.0)) / 12.0;

println("(1) Adjustable or (2) Fixed shelving? ")
opt = parse(chomp(readline()))

if (opt == 1)
   shlfSag = 5.0 * loadShlf * shlfLen^4.0
elseif (opt == 2) then
   shlfSag = loadShlf * shlfLen^4.0
end
shlfSag = shlfSag / (384.0 * modElas * momentI)

println("Shelf deflection (sag) = ", round(shlfSag,3), " inches")

if (shlfSag/(shlfLen/12.0) < 0.02)
   println("Shelf sag is acceptable")
else
   println("Shelf sag is unacceptable")
end

APPENDIX: ORIGINAL PROGRAM

program shelfLoad

   implicit none

   real :: loadFt, shlfLen, shlfDep, shlfThk
   real :: modElas, loadShlf, momentI, shlfSag
   integer :: opt

   write (*,*) "Calculate Shelf Deflection (Sag)"

   write (*,fmt="(a21)",advance="no") "Load per foot (lbs): "
   read (*,*) loadFt
   write (*,fmt="(a19)",advance="no") "Shelf length (in): "
   read (*,*) shlfLen
   write (*,fmt="(a18)",advance="no") "Shelf depth (in): "
   read (*,*) shlfDep
   write (*,fmt="(a22)",advance="no") "Shelf thickness (in): "
   read (*,*) shlfThk
   write (*,fmt="(a29)",advance="no") "Modulus of Elasticity (psi): "
   read (*,*) modElas

   ! Calculate uniform load lb/in
   loadShlf = loadFt / 12.0
   write (*,10) "Shelf load = ", loadShlf, " lb/in"
   10 format (a13, f5.2, a6)

   ! Calculate area moment of inertia
   momentI = (shlfDep * (shlfThk**3.0)) / 12.0;

   write (*,fmt="(a38)",advance="no") "(1) Adjustable or (2) Fixed shelving? 
"
   read (*,*) opt

   if (opt == 1) then
      shlfSag = 5.0 * loadShlf * shlfLen**4.0
   elseif (opt == 2) then
      shlfSag = loadShlf * shlfLen**4.0
   end if
   shlfSag = shlfSag / (384.0 * modElas * momentI)

   write (*,20) "Shelf deflection (sag) = ", shlfSag, " inches"
   20 format(a25, f5.3, a7)

   if (shlfSag/(shlfLen/12.0) < 0.02) then
      write (*,*) "Shelf sag is acceptable"
   else
      write (*,*) "Shelf sag is unacceptable"
   end if

end program shelfLoad

APPENDIX: EXTENDED PROGRAM

program shelfLoad

   implicit none

   integer, dimension(9) :: modElas = [1770000, 1490000, 1830000, &
             1780000, 1680000, 1300000, 1580000, 580000, 1480000]
   character(len=15), dimension(9) :: woods
   real :: loadFt, shlfLen, shlfDep, shlfThk
   real :: loadShlf, momentI, shlfSag
   integer :: i, opt, woodOpt

   woods = [character(15) :: 'Ash','Cherry','Hard Maple','White Oak',&
            'Walnut','Pine','Poplar','MDF','Birch Plywood']

   write (*,*) "Calculate Shelf Deflection (Sag)"

   write (*,fmt="(a21)",advance="no") "Load per foot (lbs): "
   read (*,*) loadFt
   write (*,fmt="(a19)",advance="no") "Shelf length (in): "
   read (*,*) shlfLen
   write (*,fmt="(a18)",advance="no") "Shelf depth (in): "
   read (*,*) shlfDep
   write (*,fmt="(a22)",advance="no") "Shelf thickness (in): "
   read (*,*) shlfThk

   ! Calculate uniform load lb/in
   loadShlf = loadFt / 12.0
   write (*,10) "Shelf load = ", loadShlf, " lb/in"
   10 format (a13, f5.2, a6)

   write (*,fmt="(a23)") "Choose a wood species: "
   do i= 1,9
      write (*,fmt="(a3,i1,a2,a15)") "  (",i,") ",woods(i)
   end do
   write (*,fmt="(a2)",advance="no") "? "
   read (*,*) woodOpt

   ! Calculate area moment of inertia
   momentI = (shlfDep * (shlfThk**3.0)) / 12.0;

   write (*,fmt="(a38)",advance="no") "(1) Adjustable or (2) Fixed shelving? 
"
   read (*,*) opt

   if (opt == 1) then
      shlfSag = 5.0 * loadShlf * shlfLen**4.0
   elseif (opt == 2) then
      shlfSag = loadShlf * shlfLen**4.0
   end if
   shlfSag = shlfSag / (384.0 * modElas(woodOpt) * momentI)

   write (*,20) "Shelf deflection (sag) = ", shlfSag, " inches"
   20 format(a25, f5.3, a7)

   if (shlfSag/(shlfLen/12.0) < 0.02) then
      write (*,*) "Shelf sag is acceptable"
   else
      write (*,*) "Shelf sag is unacceptable"
   end if

end program shelfLoad

